import crafttweaker.api.item.IItemStack;
/*红石*/
<recipetype:minecraft:smelting>.removeRecipe(<item:minecraft:redstone>);//从熔炉移除配方
<recipetype:minecraft:blasting>.removeRecipe(<item:minecraft:redstone>);//从高炉移除配方
<recipetype:create:mixing>.addRecipe("meng_red_stone",
	"none",
	<item:minecraft:redstone>*5,
	[<item:minecraft:red_dye>,<item:appliedenergistics2:nether_quartz_dust>]
);

/*石头*/
recipes.addJSONRecipe("meng_stone",{
	type:"minecraft:campfire_cooking",
	ingredient:{
		item:<item:minecraft:bone_block>.registryName
	},
	result:<item:minecraft:stone>,
	experience: 0.30,
	cookingtime: 100
});

val minecraft = new Make();
/*骨块*/
minecraft.compound("meng_bone_block",<item:minecraft:bone_block>,<item:minecraft:bone>,<item:minecraft:bone>);

/*熔炉*/
minecraft.compound("meng_furnace",<item:minecraft:furnace>,
					<item:minecraft:stone>,<item:minecraft:cobblestone>,<item:minecraft:stone>,
					<item:minecraft:cobblestone>,<item:minecraft:air>,<item:minecraft:cobblestone>,
					<item:minecraft:stone>,<item:minecraft:iron_ingot>,<item:minecraft:stone>);

/*闪长岩*/
recipes.removeRecipe(<item:minecraft:diorite>);
craftingTable.addShaped("meng_diorite",<item:minecraft:diorite>*2,[
	[<tag:items:forge:cobblestone>,<item:minecraft:white_dye>],
	[<item:minecraft:white_dye>,<tag:items:forge:cobblestone>]
]);

/*岩浆桶*/
<recipetype:create:mixing>.addRecipe("meng_lava_bucket",
	"heated",
	<item:minecraft:lava_bucket>*1,
	[<item:minecraft:water_bucket>,<item:minecraft:redstone_block>]
);

/*沙砾*/
<recipetype:create:compacting>.addRecipe("meng_gravel",
	"none",
	<item:minecraft:gravel>*1,
	[<item:minecraft:cobblestone> * 2]
);

/*沙子*/
<recipetype:create:compacting>.addRecipe("meng_sand",
	"heated",
	<item:minecraft:sand>*1,
	[<item:minecraft:gravel> * 2]
);

/*灵魂沙*/
<recipetype:create:mixing>.addRecipe("meng_soul_sand",
	"heated",
	<item:minecraft:soul_sand>*1,
	[<item:minecraft:egg> * 3,<item:minecraft:sand>]
);

var MengOreName as string[] = [
	"meng_iron_ore",//铁
	"meng_coal_ore",//煤炭
	"meng_gold_ore",//金
	"meng_lapis_ore",//青金石
	"meng_redstone_ore",//红石
	"meng_aluminum_ore",//铝
	"meng_silver_ore",//银
	"meng_lead_ore",//铅
	"meng_nickel_ore",//镍

];

var RawOre as IItemStack[] = [
	<item:minecraft:iron_ore>,//铁矿
	<item:minecraft:iron_nugget>,//铁粒
	<item:minecraft:coal_ore>,//煤炭矿
	<item:morecharcoal:coal_chunk>,//煤粒
	<item:minecraft:gold_ore>,//金矿
	<item:minecraft:gold_nugget>,//金粒
	<item:minecraft:lapis_ore>,//青金石矿
	<item:minecraft:lapis_lazuli>,//青金石
	<item:minecraft:redstone_ore>,//红石矿
	<item:minecraft:redstone>,//红石
	<item:immersiveengineering:ore_aluminum>,//铝土矿
	<item:immersiveengineering:nugget_aluminum>,//铝粒
	<item:immersiveengineering:ore_silver>,//银矿石
	<item:immersiveengineering:nugget_silver>,
	<item:immersiveengineering:ore_lead>,//铅矿石
	<item:immersiveengineering:nugget_lead>,//
	<item:immersiveengineering:ore_nickel>,//镍矿
	<item:immersiveengineering:nugget_nickel>,
];


val makeOre1 = new Make();
var jishu as int = 0;
for i in 0 .. MengOreName.length{
	makeOre1.buildOre(MengOreName[i],RawOre[jishu],RawOre[jishu+1]);
	jishu+=2;
}



